/**
*提示有错误的对话框,点关闭就关闭，还可以发一个事件给用户
*mc_dbox里有一个tf_tip,btn_ok,btn_cancel//其中bnt_canel可以不需要，点他不会向外部发送事件，只会关闭
*mc_mb是一个很大的鼠标屏蔽，能遮住整个
*/

package kgame5.kgui.kcomp {
import kgame5.kutil.*;

import flash.display.*;
import flash.events.*;
import flash.text.*;

public class SmartAlarm extends EventDispatcher {
	public static const 
		EVENTTYPE_OK:String="EVENTTYPE_OK",//用户点了确定
		EVENTTYPE_CANCEL:String="EVENTTYPE_CANCEL";//用户点了取消

	//___________________Property____________________
	
	//--总模型
	private var 
		mc:MovieClip,
			mc_dbox : MovieClip,
				tf_tip:TextField,
				btn_ok:SimpleButton,
				btn_cancel:SimpleButton,
			mc_mb:MovieClip;//
	
	//___________________Method_____________________

	public function SmartAlarm(p_mc:MovieClip) {
		////T.p("MyFaceTimerover: "+p_mc);
		mc=p_mc;
			mc_dbox=mc.mc_dbox;
				tf_tip=mc_dbox.tf_tip;
				btn_ok=mc_dbox.btn_ok;
				btn_cancel=mc_dbox.btn_cancel;
				
			mc_mb=mc.mc_mb;
		
		Util.OnClick(btn_ok,onClickOK);
		if(btn_cancel!=null){
			Util.OnClick(btn_cancel,onClickCancel);
		}
		
		closePane();
	}
	//可能要做些其他的操作要用到这个
	public function getDBox():MovieClip{
		return mc_dbox;
	}
	private function onClickOK(e:MouseEvent){
		closePane();
		this.dispatchEvent(new KEvent(EVENTTYPE_OK));
		
	}
	private function onClickCancel(e:MouseEvent){
		closePane();
		this.dispatchEvent(new KEvent(EVENTTYPE_CANCEL));
	}
	
	///////////////////打开或者关闭
	public function openPane(sTip:String=null){
		mc.visible=true;
		if(sTip!=null){
			tf_tip.text=sTip;
		}
	}
	public function closePane(){
		mc.visible=false;
	}
} //end of class
}